Rules for integrands involving piecewise linear functions

1: $\int u^m \, dx$ when $\partial_x u == c$

Derivation: Integration by substitution

Basis: If $\partial_x u == c$, then $u^m == \frac{1}{c} u^m \partial_x u$

Rule: If $\partial_x u == c$, then

$$\int u^m \, dx \longrightarrow \frac{1}{c} \text{Subst}\left[\int x^m \, dx, x, u\right]$$

Program code:

```
Int[u_^m_.,x_Symbol] :=
  With[{c=Simplify[D[u,x]]},
  1/c*Subst[Int[x^m,x],x,u]] /;
FreeQ[m,x] && PiecewiseLinearQ[u,x]
```

2: $\int u^m v^n \, dx$ when $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0$

1. $\int \frac{v^n}{u} \, dx$ when $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0$

1. $\int \frac{v^n}{u} \, dx$ when $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0 \land n > 0$

1: $\int \frac{v}{u} \, dx$ when $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0$

Derivation: Piecewise linear recurrence 2 with $m = -1$ and $n = 1$

Derivation: Inverted integration by parts

– Rule: If $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0$, then

$$\int \frac{v}{u} \, dx \rightarrow \frac{b x}{a} - \frac{b u - a v}{a} \int \frac{1}{u} \, dx$$

– Program code:

```
Int[v_/u_,x_Symbol] :=
  With[{a=Simplify[D[u,x]],b=Simplify[D[v,x]]},
  b*x/a - (b*u-a*v)/a*Int[1/u,x] /;
 NeQ[b*u-a*v,0]] /;
PiecewiseLinearQ[u,v,x]
```

2: $\int \frac{v^n}{u}\,dx$ when $\partial_x u = a \land \partial_x v = b \land b u - a v \neq 0 \land n > 0 \land n \neq 1$

Derivation: Piecewise linear recurrence 2 with $m = -1$

Derivation: Inverted integration by parts

Rule: If $\partial_x u = a \land \partial_x v = b \land b u - a v \neq 0 \land n > 0 \land n \neq 1$, then

$$\int \frac{v^n}{u}\,dx \longrightarrow \frac{v^n}{a n} - \frac{b u - a v}{a} \int \frac{v^{n-1}}{u}\,dx$$

Program code:

```
Int[v_^n_/u_,x_Symbol] :=
  With[{a=Simplify[D[u,x]],b=Simplify[D[v,x]]},
  v^n/(a*n) - (b*u-a*v)/a*Int[v^(n-1)/u,x] /;
 NeQ[b*u-a*v,0]] /;
PiecewiseLinearQ[u,v,x] && GtQ[n,0] && NeQ[n,1]
```

2. $\int \frac{v^n}{u}\,dx$ when $\partial_x u == a \land \partial_x v == b \land b\,u - a\,v \neq 0 \land n < 0$

1: $\int \frac{1}{u\,v}\,dx$ when $\partial_x u == a \land \partial_x v == b \land b\,u - a\,v \neq 0$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: $\frac{1}{u\,v} == \frac{b}{b\,u-a\,v}\,\frac{1}{v} - \frac{a}{b\,u-a\,v}\,\frac{1}{u}$

Basis: If $\partial_x u == a \land \partial_x v == b \land b\,u - a\,v \neq 0$, then $\partial_x \frac{1}{b\,u-a\,v} == 0$

Rule: If $\partial_x u == a \land \partial_x v == b \land b\,u - a\,v \neq 0$, then

$$\int \frac{1}{u\,v}\,dx \longrightarrow \frac{b}{b\,u-a\,v}\int \frac{1}{v}\,dx - \frac{a}{b\,u-a\,v}\int \frac{1}{u}\,dx$$

Program code:

```
Int[1/(u_*v_),x_Symbol] :=
  With[{a=Simplify[D[u,x]],b=Simplify[D[v,x]]},
  b/(b*u-a*v)*Int[1/v,x] - a/(b*u-a*v)*Int[1/u,x] /;
 NeQ[b*u-a*v,0]] /;
PiecewiseLinearQ[u,v,x]
```

2. $\int \frac{1}{u\sqrt{v}}\,dx$ when $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0$

1: $\int \frac{1}{u\sqrt{v}}\,dx$ when $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0 \land \frac{b u - a v}{a} > 0$

–

Rule: If $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0 \land \frac{b u - a v}{a} > 0$, then

$$\int \frac{1}{u\sqrt{v}}\,dx \to \frac{2}{a\sqrt{\frac{b u - a v}{a}}}\,\text{ArcTan}\left[\frac{\sqrt{v}}{\sqrt{\frac{b u - a v}{a}}}\right]$$

–

Program code:

```
Int[1/(u_*Sqrt[v_]),x_Symbol] :=
  With[{a=Simplify[D[u,x]],b=Simplify[D[v,x]]},
  2*ArcTan[Sqrt[v]/Rt[(b*u-a*v)/a,2]]/(a*Rt[(b*u-a*v)/a,2]) /;
 NeQ[b*u-a*v,0] && PosQ[(b*u-a*v)/a]] /;
PiecewiseLinearQ[u,v,x]
```

2: $\int \frac{1}{u\sqrt{v}}\,dx$ when $\partial_x u == a \land \partial_x v == b \land b\,u - a\,v \neq 0 \land \neg\left(\frac{b\,u-a\,v}{a} > 0\right)$

- Rule: If $\partial_x u == a \land \partial_x v == b \land b\,u - a\,v \neq 0 \land \neg\left(\frac{b\,u-a\,v}{a} > 0\right)$, then

$$\int \frac{1}{u\sqrt{v}}\,dx \longrightarrow -\frac{2}{a\sqrt{-\frac{b\,u-a\,v}{a}}}\,\text{ArcTanh}\left[\frac{\sqrt{v}}{\sqrt{-\frac{b\,u-a\,v}{a}}}\right]$$

- Program code:

```
Int[1/(u_*Sqrt[v_]),x_Symbol] :=
  With[{a=Simplify[D[u,x]],b=Simplify[D[v,x]]},
  -2*ArcTanh[Sqrt[v]/Rt[-(b*u-a*v)/a,2]]/(a*Rt[-(b*u-a*v)/a,2]) /;
 NeQ[b*u-a*v,0] && NegQ[(b*u-a*v)/a]] /;
PiecewiseLinearQ[u,v,x]
```

3: $\int \frac{v^n}{u}\, dx$ when $\partial_x u == a \land \partial_x v == b \land b\,u - a\,v \neq 0 \land n < -1$

Derivation: Piecewise linear recurrence 3 with $n = -1$

Derivation: Integration by parts

Rule: If $\partial_x u == a \land \partial_x v == b \land b\,u - a\,v \neq 0 \land n < -1$, then

$$\int \frac{v^n}{u}\, dx \longrightarrow \frac{v^{n+1}}{(n+1)\,(b\,u - a\,v)} - \frac{a\,(n+1)}{(n+1)\,(b\,u - a\,v)} \int \frac{v^{n+1}}{u}\, dx$$

Program code:

```
Int[v_^n_/u_,x_Symbol] :=
  With[{a=Simplify[D[u,x]],b=Simplify[D[v,x]]},
  v^(n+1)/((n+1)*(b*u-a*v)) -
   a*(n+1)/((n+1)*(b*u-a*v))*Int[v^(n+1)/u,x] /;
 NeQ[b*u-a*v,0]] /;
PiecewiseLinearQ[u,v,x] && LtQ[n,-1]
```

3: $\int \frac{v^n}{u} \, dx$ when $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0 \land n \notin \mathbb{Z}$

Rule: If $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0 \land n \notin \mathbb{Z}$, then

$$\int \frac{v^n}{u} \, dx \longrightarrow \frac{v^{n+1}}{(n+1)\,(b u - a v)} \text{Hypergeometric2F1}\left[1,\, n+1,\, n+2,\, -\frac{a v}{b u - a v}\right]$$

Program code:

```
Int[v_^n_/u_,x_Symbol] :=
  With[{a=Simplify[D[u,x]],b=Simplify[D[v,x]]},
  v^(n+1)/((n+1)*(b*u-a*v))*Hypergeometric2F1[1,n+1,n+2,-a*v/(b*u-a*v)] /;
 NeQ[b*u-a*v,0]] /;
PiecewiseLinearQ[u,v,x] && Not[IntegerQ[n]]
```

2. $\int \frac{1}{\sqrt{u}\,\sqrt{v}} \, dx$ when $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0$

1: $\int \frac{1}{\sqrt{u}\,\sqrt{v}} \, dx$ when $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0 \land a b > 0$

Rule: If $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0 \land a b > 0$, then

$$\int \frac{1}{\sqrt{u}\,\sqrt{v}} \, dx \longrightarrow \frac{2}{\sqrt{a b}} \text{ArcTanh}\left[\frac{\sqrt{a b}\,\sqrt{u}}{a \sqrt{v}}\right]$$

Program code:

```
Int[1/(Sqrt[u_]*Sqrt[v_]),x_Symbol] :=
  With[{a=Simplify[D[u,x]],b=Simplify[D[v,x]]},
  2/Rt[a*b,2]*ArcTanh[Rt[a*b,2]*Sqrt[u]/(a*Sqrt[v])] /;
 NeQ[b*u-a*v,0] && PosQ[a*b]] /;
PiecewiseLinearQ[u,v,x]
```

2: $\int \frac{1}{\sqrt{u}\,\sqrt{v}}\,dx$ when $\partial_x u == a \land \partial_x v == b \land b\,u - a\,v \neq 0 \land \neg\,(a\,b > 0)$

Rule: If $\partial_x u == a \land \partial_x v == b \land b\,u - a\,v \neq 0 \land \neg\,(a\,b > 0)$, then

$$\int \frac{1}{\sqrt{u}\,\sqrt{v}}\,dx \longrightarrow \frac{2}{\sqrt{-a\,b}}\,\text{ArcTan}\left[\frac{\sqrt{-a\,b}\,\sqrt{u}}{a\,\sqrt{v}}\right]$$

Program code:

```
Int[1/(Sqrt[u_]*Sqrt[v_]),x_Symbol] :=
  With[{a=Simplify[D[u,x]],b=Simplify[D[v,x]]},
  2/Rt[-a*b,2]*ArcTan[Rt[-a*b,2]*Sqrt[u]/(a*Sqrt[v])] /;
 NeQ[b*u-a*v,0] && NegQ[a*b]] /;
PiecewiseLinearQ[u,v,x]
```

3: $\int u^m\,v^n\,dx$ when $\partial_x u == a \land \partial_x v == b \land b\,u - a\,v \neq 0 \land m + n + 2 == 0 \land m \neq -1$

Derivation: Piecewise linear recurrence 3 with $m + n + 2 == 0$

Rule: If $\partial_x u == a \land \partial_x v == b \land b\,u - a\,v \neq 0 \land m + n + 2 == 0 \land m \neq -1$, then

$$\int u^m\,v^n\,dx \longrightarrow -\frac{u^{m+1}\,v^{n+1}}{(m+1)\,(b\,u - a\,v)}$$

Program code:

```
Int[u_^m_*v_^n_,x_Symbol] :=
  With[{a=Simplify[D[u,x]],b=Simplify[D[v,x]]},
  -u^(m+1)*v^(n+1)/((m+1)*(b*u-a*v)) /;
 NeQ[b*u-a*v,0]] /;
FreeQ[{m,n},x] && PiecewiseLinearQ[u,v,x] && EqQ[m+n+2,0] && NeQ[m,-1]
```

4: $\int u^m v^n \, dx$ when $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0 \land m < -1 \land n > 0$

Derivation: Piecewise linear recurrence 1

Derivation: Integration by parts

Rule: If $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0 \land m + n + 2 \neq 0 \land m < -1 \land n > 0$, then

$$\int u^m v^n \, dx \longrightarrow \frac{u^{m+1} v^n}{a (m+1)} - \frac{b n}{a (m+1)} \int u^{m+1} v^{n-1} \, dx$$

Program code:

```
Int[u_^m_*v_^n_.,x_Symbol] :=
  With[{a=Simplify[D[u,x]],b=Simplify[D[v,x]]},
  u^(m+1)*v^n/(a*(m+1)) -
   b*n/(a*(m+1))*Int[u^(m+1)*v^(n-1),x] /;
 NeQ[b*u-a*v,0]] /;
FreeQ[{m,n},x] && PiecewiseLinearQ[u,v,x] (* && NeQ[m+n+2,0] *) && NeQ[m,-1] && (
  LtQ[m,-1] && GtQ[n,0] && Not[ILtQ[m+n,-2] && (FractionQ[m] || GeQ[2*n+m+1,0])] ||
  IGtQ[n,0] && IGtQ[m,0] && LeQ[n,m] ||
(* ILtQ[n,0] && ILtQ[m,0] && LeQ[n,m] || *)
  IGtQ[n,0] && Not[IntegerQ[m]] ||
  ILtQ[m,0] && Not[IntegerQ[n]])
```

5: $\int u^m v^n \, dx$ when $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0 \land m + n + 2 \neq 0 \land n > 0 \land m + n + 1 \neq 0$

Derivation: Piecewise linear recurrence 2

Derivation: Inverted integration by parts

Rule: If $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0 \land m + n + 2 \neq 0 \land n > 0 \land m + n + 1 \neq 0$, then

$$\int u^m v^n \, dx \longrightarrow \frac{u^{m+1} v^n}{a (m+n+1)} - \frac{n (b u - a v)}{a (m+n+1)} \int u^m v^{n-1} \, dx$$

–

Program code:

```
Int[u_^m_*v_^n_.,x_Symbol] :=
  With[{a=Simplify[D[u,x]],b=Simplify[D[v,x]]},
  u^(m+1)*v^n/(a*(m+n+1)) -
   n*(b*u-a*v)/(a*(m+n+1))*Int[u^m*v^(n-1),x] /;
 NeQ[b*u-a*v,0]] /;
PiecewiseLinearQ[u,v,x] && NeQ[m+n+2,0] && GtQ[n,0] && NeQ[m+n+1,0] &&
  Not[IGtQ[m,0] && (Not[IntegerQ[n]] || LtQ[0,m,n])] &&
  Not[ILtQ[m+n,-2]]
```

```
Int[u_^m_*v_^n_,x_Symbol] :=
  With[{a=Simplify[D[u,x]],b=Simplify[D[v,x]]},
  u^(m+1)*v^n/(a*(m+n+1)) -
   n*(b*u-a*v)/(a*(m+n+1))*Int[u^m*v^Simplify[n-1],x] /;
 NeQ[b*u-a*v,0]] /;
PiecewiseLinearQ[u,v,x] && NeQ[m+n+1,0] && Not[RationalQ[n]] && SumSimplerQ[n,-1]
```

6: $\int u^m v^n \, dx$ when $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0 \land m + n + 2 \neq 0 \land m < -1$

Derivation: Piecewise linear recurrence 3

Derivation: Integration by parts

Basis: $u^m v^n == v^{m+n+2} \frac{u^m}{v^{m+2}}$

Rule: If $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0 \land m + n + 2 \neq 0 \land m < -1$, then

$$\int u^m v^n \, dx \longrightarrow -\frac{u^{m+1} v^{n+1}}{(m+1)(b u - a v)} + \frac{b (m+n+2)}{(m+1)(b u - a v)} \int u^{m+1} v^n \, dx$$

Program code:

```
Int[u_^m_*v_^n_,x_Symbol] :=
  With[{a=Simplify[D[u,x]],b=Simplify[D[v,x]]},
  -u^(m+1)*v^(n+1)/((m+1)*(b*u-a*v)) +
   b*(m+n+2)/((m+1)*(b*u-a*v))*Int[u^(m+1)*v^n,x] /;
 NeQ[b*u-a*v,0]] /;
PiecewiseLinearQ[u,v,x] && NeQ[m+n+2,0] && LtQ[m,-1]
```

```
Int[u_^m_*v_^n_,x_Symbol] :=
  With[{a=Simplify[D[u,x]],b=Simplify[D[v,x]]},
  -u^(m+1)*v^(n+1)/((m+1)*(b*u-a*v)) +
   b*(m+n+2)/((m+1)*(b*u-a*v))*Int[u^Simplify[m+1]*v^n,x] /;
 NeQ[b*u-a*v,0]] /;
PiecewiseLinearQ[u,v,x] && Not[RationalQ[m]] && SumSimplerQ[m,1]
```

7: $\int u^m v^n \, dx$ when $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0 \land m \notin \mathbb{Z} \land n \notin \mathbb{Z}$

Rule: If $\partial_x u == a \land \partial_x v == b \land b u - a v \neq 0 \land m \notin \mathbb{Z} \land n \notin \mathbb{Z}$, then

$$\int u^m v^n \, dx \longrightarrow \frac{u^m v^{n+1}}{b\,(n+1)\,\left(\frac{b\,u}{b\,u-a\,v}\right)^m} \text{Hypergeometric2F1}\left[-m,\, n+1,\, n+2,\, -\frac{a\,v}{b\,u-a\,v}\right]$$

Program code:

```
Int[u_^m_*v_^n_,x_Symbol] :=
  With[{a=Simplify[D[u,x]],b=Simplify[D[v,x]]},
  u^m*v^(n+1)/(b*(n+1)*(b*u/(b*u-a*v))^m)*Hypergeometric2F1[-m,n+1,n+2,-a*v/(b*u-a*v)] /;
 NeQ[b*u-a*v,0]] /;
PiecewiseLinearQ[u,v,x] && Not[IntegerQ[m]] && Not[IntegerQ[n]]
```

3. $\int u^n (a+b\,x)^m \, \text{Log}[a+b\,x] \, dx$ when $\partial_x u == c$

1: $\int u^n \, \text{Log}[a+b\,x] \, dx$ when $\partial_x u == c \wedge n > 0$

Derivation: Integration by parts

Basis: If $\partial_x u == c$, then $\partial_x \left(u^n \, \text{Log}[a+b\,x]\right) == \frac{b\,u^n}{a+b\,x} + c\,n\,u^{n-1} \, \text{Log}[a+b\,x]$

Rule: If $\partial_x u == c \wedge n > 0$, then

$$\int u^n \, \text{Log}[a+b\,x] \, dx \longrightarrow \frac{u^n\,(a+b\,x)\,\text{Log}[a+b\,x]}{b} - \int u^n \, dx - \frac{c\,n}{b} \int u^{n-1}\,(a+b\,x)\,\text{Log}[a+b\,x] \, dx$$

Program code:

```
Int[u_^n_.*Log[a_.+b_.*x_],x_Symbol] :=
  With[{c=Simplify[D[u,x]]},
  u^n*(a+b*x)*Log[a+b*x]/b -
  Int[u^n,x] -
  c*n/b*Int[u^(n-1)*(a+b*x)*Log[a+b*x],x]] /;
FreeQ[{a,b},x] && PiecewiseLinearQ[u,x] && Not[LinearQ[u,x]] && GtQ[n,0]
```

2. $\int u^n\,(a+b\,x)^m\,\mathrm{Log}[a+b\,x]\,dx$ when $\partial_x u == c$

x: $\int \frac{u^n\,\mathrm{Log}[a+b\,x]}{a+b\,x}\,dx$ when $\partial_x u == c \wedge n > 0$

Derivation: Integration by parts with a double-back flip

Basis: If $\partial_x u == c$, then $\partial_x \left(u^n\,\mathrm{Log}[a+b\,x]\right) == \frac{b\,u^n}{a+b\,x} + c\,n\,u^{n-1}\,\mathrm{Log}[a+b\,x]$

Rule: If $\partial_x u == c \wedge n > 0$, then

$$\int \frac{u^n\,\mathrm{Log}[a+b\,x]}{a+b\,x}\,dx \longrightarrow \frac{u^n\,\mathrm{Log}[a+b\,x]^2}{2\,b} - \frac{c\,n}{2\,b}\int u^{n-1}\,\mathrm{Log}[a+b\,x]^2\,dx$$

Program code:

```
(* Int[u_^n_.*Log[a_.+b_.*x_]/(a_.+b_.*x_),x_Symbol] :=
  With[{c=Simplify[D[u,x]]},
  u^n*Log[a+b*x]^2/(2*b) -
  c*n/(2*b)*Int[u^(n-1)*Log[a+b*x]^2,x]] /;
FreeQ[{a,b},x] && PiecewiseLinearQ[u,x] && GtQ[n,0] *)
```

2: $\int u^n (a+b x)^m \operatorname{Log}[a+b x] \, dx$ when $\partial_x u == c \land n>0 \land m \neq -1$

Derivation: Integration by parts

– Basis: If $\partial_x u == c$, then $\partial_x \left(u^n \operatorname{Log}[a+b x]\right) == \frac{b u^n}{a+b x} + c n u^{n-1} \operatorname{Log}[a+b x]$

Rule: If $\partial_x u == c \land n>0 \land m \neq -1$, then

$$\int u^n (a+b x)^m \operatorname{Log}[a+b x] \, dx \longrightarrow$$
$$\frac{u^n (a+b x)^{m+1} \operatorname{Log}[a+b x]}{b (m+1)} - \frac{1}{m+1} \int u^n (a+b x)^m \, dx - \frac{c n}{b (m+1)} \int u^{n-1} (a+b x)^{m+1} \operatorname{Log}[a+b x] \, dx$$

Program code:

```
Int[u_^n_.*(a_.+b_.*x_)^m_.*Log[a_.+b_.*x_],x_Symbol] :=
  With[{c=Simplify[D[u,x]]},
  u^n*(a+b*x)^(m+1)*Log[a+b*x]/(b*(m+1)) -
  1/(m+1)*Int[u^n*(a+b*x)^m,x] -
  c*n/(b*(m+1))*Int[u^(n-1)*(a+b*x)^(m+1)*Log[a+b*x],x]] /;
FreeQ[{a,b,m},x] && PiecewiseLinearQ[u,x] && Not[LinearQ[u,x]] && GtQ[n,0] && NeQ[m,-1]
```